
Abstract
Medication adherence measures the extent to which a patient takes medications as prescribed by their health care
provider. Published studies note that up to half of chronically ill patients do not adhere to their medication regimens.
The financial impact of non-adherence is estimated at $100 billion annually due to avoidable hospitalizations with
other health factors adding to that cost. To increase medication adherence and decrease overall healthcare spend, it
is vital that health care providers be able to accurately measure adherence and identify potential gaps in care. In this
paper, we will demonstrate SAS®-based solutions which allow providers to calculate adherence across a range of
prescribing patterns. The code provided allows PDC to be calculated at both the therapeutic class level and the
patient disease level. Refining existing methodologies has increased the efficiency of the calculations.

Introduction
There are numerous methods for calculating adherence; however, researchers favor the use of the Medication
Possession Ratio (MPR) or the Proportion of Days Covered (PDC). Generally, MPR measures adherence by
comparing the days of medication supply to the number of days in a time interval (Steiner, 1997). It should be noted
that MPR has been criticized as of late for overestimating the true rate of adherence. This is true as MPR is the
simple summation of “days of supply”. The calculation for PDC measures the number of days “covered” by a
prescription and divides by the number of days in the measurement period. The PDC looks at each day in the interval
to see if the patient has the prescribed drug on-hand. Due to results from recent studies of the two methods, PDC has
become the preferred methodology. The Pharmacy Quality Alliance (PQA) published a review of the two calculations
in which they proclaimed PDC as their “preferred method” of adherence calculation.
This paper presents two macros for calculating PDC, one to be used if a patient maintains a single drug therapy
throughout the measurement period (%PDC_Single) and another to be used in the event a patient switches between
equivalent therapies during the course of treatment (%PDC_Change). %PDC_Single uses a DO loop as the primary
means of calculation for a single drug therapy. %PDC_Change handles drug therapy switching via a data step in the
macro. The steps in %PDC_Change are more complex than %PDC_Single, but they both are intended to offset
known problems with the MPR calculation as noted previously.
All medication utilization measured by either of these macros depends on the drugs being identified via Medi-Span’s
Generic Product Identifier (GPI) number. Specifically, we use the six-digit form (known as “GPI6”) to assure that
drugs are pharmaceutically equivalent with regard to active ingredients.

1: Measuring PDC for Single Drug Therapy
Patient A receives prescriptions for a cholesterol-lowering drug. The record of her prescription fills are listed in
Table1. As all of the fills have the same GPI6 identifier, the patient is considered to be on single drug therapy.
Table 1: Prescription Fills – Patient A
Fill Date

Days of
Supply

394000---HMG-CoA Reductase Inhibitors

12/20/2010

30

A

394000---HMG-CoA Reductase Inhibitors

08/25/2011

30

A

394000---HMG-CoA Reductase Inhibitors

09/05/2011

30

A

394000---HMG-CoA Reductase Inhibitors

09/26/2011

30

A

394000---HMG-CoA Reductase Inhibitors

12/10/2011

30

Patient ID

GPI6

A

For the purpose of this example, the study period is calendar year 2011. As the start date of the measurement period
is January 1, 2011, any prescription fills prior to that date are excluded from the calculation, even though the impact
of such a fill may extend into the measurement period. Thus, the 30-day supply filled on 12/20/2010 is excluded from
the calculation. Therefore, in Figure 2, the patient is shown to have a zero quantity of the drug for the first eighteen
days of 2011.
Prior to running the macro %PDC_Single, there are seven parameters that must be defined:

Input,

Output,

Patient identifier (PID),

Prescription fill date (fill_dt),

Days of supply per fill (dos),

Study period beginning date (start_dt), and

Study period ending date (end_dt)
Before the fill dates can be analyzed and adjusted, we must define the study period, set values for the macro
variables, and create arrays to reflect the dates and days of supply for each fill. These initial steps in the macro
prepare the data so that the fill dates can be adjusted as necessary.
Once the preceding steps are complete (see annotations in SAS code below), the macro will adjust the patient’s Rx
fill dates to account for early refills. For example, the patient fills a 30-day supply of the drug on 8/25/2011. That 30day supply will be finished on 9/23/2011. However, the patient refills early on 9/5/2011. In order to avoid any overlap
and properly note that the patient has adequate medication coverage, her refill date of 9/5/2011 is adjusted to
9/24/2011. The resulting change is the variable labeled “adj_fill_date”.

Following this step, we truncate any medication supply that falls outside of the study period. As shown in Table 1, the
patient filled a 30-day prescription on 12/10/2011. The medication should be used up by 1/8/2012; however, our study
period ends at 12/31/2011. As such, the 30-day fill is changed to reflect a 22-day supply.
The final step in the macro is to sum the adjusted days of supply to calculate the individual patient’s PDC. In this
example, the PDC for this patient in calendar year 2011 is as follows:

sum of days covered / total days in study period

112 / 129 =86.82 % (129 is the number of days between her initial fill date of 8/25/2011 and the end date of
the study period, 12/31/2011)
The 86.82% shown above is the PDC calculated for a single patient using only one drug therapy. The implementation
of a “group by” clause would allow for calculating PDC at additional levels, including patient disease level or by
therapeutic class.
****************************************************************************
** Macro Parameters:
***
** input
- Input Dataset name
***
** output
- Output Dataset name
***
** pid
- Unique patient identifier
***
** fill_dt
- Prescription fill date
***
** dos
- Fill days of supply
***
** start_dt
- Start date of study period
***
** end_dt
- End date of study period
***
****************************************************************************
%macro PDC_Single(
input=,
output=,
pid=,
fill_dt=,
dos=,
start_dt=,
end_dt=);
** Step 1: Exclude fills outside of study period **;
data &input.;
set &input.;
if &end_dt. >= &fill_dt >= &start_dt.;
run;
** Step 2: Find out the maximum number of fills a patient can have **;
**
and save it into macro variable, &_mcount
**;
proc sql noprint;
create table temp as
select &pid., count(*) as ct
from &input.
group by &pid.
;
select max(ct) into :_mcount from temp;
quit;
%let _mcount=&_mcount;
** Step 3: Transpose dataset **;
proc transpose data=&input. out=fill_dates(drop=_name_) prefix=fill_dt;
by &pid.;
var &fill_dt.;
run;
proc transpose data=&input. out=fill_DOS(drop=_name_) prefix=days_sply;
by &pid.;
var &dos.;
run;
data claims_refmt;
merge fill_dates fill_DOS;
by &pid.;
start_study_dt=fill_dt1;
end_study_dt=&end_dt.;

format start_study_dt end_study_dt mmddyy10.;
temp=end_study_dt - start_study_dt+1;
run;
data _null_;
set claims_refmt(keep=temp);
call symput('days',temp);
;
run;
** Step 4: Adjust prescription fill date for early refills **;
data pdc1(keep=&pid. adj_fill_date days_supply end_study_dt);
set claims_refmt;
array dates(*) fill_dt1 - fill_dt&_mcount.;
array dos(*) days_sply1 - days_sply&_mcount.;
adj_fill_date=dates(1);
days_supply=dos(1);
output;
do i = 2 to dim(dates) while (dates(i) ne .);
if dates(i) <dates(i-1)+dos(i-1)
then do;
dates(i) =dates(i-1)+dos(i-1);
adj_fill_date=dates(i);
days_supply=dos(i);
end;
adj_fill_date=dates(i);
days_supply=dos(i);
output;
end;
format adj_fill_date mmddyy10.;
run;
** Step 5: Truncate fills fall outside of study period **;
data pdc2;
set pdc1;
if adj_fill_date <=end_study_dt then do;
if adj_fill_date + days_supply >end_study_dt+1 then do;
days_supply=end_study_dt-adj_fill_date+1;
end;
output;
end;
run;
** Step 6: Calculate PDC at patient level
**;
proc sql;
create table &output. as
select &pid.,
sum(days_supply) as dayscovered,
(calculated dayscovered)/&days. as pdc
from pdc2
group by &pid.
;
quit;
%mend PDC_Single;
** sample macro call **;
%PDC_single(
input=claims,
output=pdc,
pid=patient_id,
fill_dt=fill_date,
dos=days_supply,
start_dt='01JAN2011'd,
end_dt='31DEC2011'd);

2: Measuring PDC When Switching Drug Therapy
If a patient is not responding to therapy, a physician may prescribe a new medication. When a patient changes drug
therapy during a measurement period, the calculation for PDC becomes more complex. Table 2 lists the prescription
drug fills for Patient B. During the measurement period, this patient took two different medications to treat symptoms
of hypertension, Loop Diuretics and Calcium Channel Blockers.
Table2. Prescription Fills – Patient B
Fill Date

Days of
Supply

372000---Loop Diuretics

12/26/2010

10

B

372000---Loop Diuretics

01/25/2011

90

B

372000---Loop Diuretics

04/10/2011

90

B

372000---Loop Diuretics

07/12/2011

30

B

340000---Calcium Channel Blockers

07/23/2011

90

B

340000---Calcium Channel Blockers

09/13/2011

90

B

372000---Loop Diuretics

11/29/2011

30

B

372000---Loop Diuretics

12/28/2011

30

Patient ID

GPI6

B

As discussed, the macro %PDC_Single adjusts a new prescription start date to coincide with the day the current fill
expires. To account for early refills, any days of supply that overlap between the two fills are then added to the new
prescription start date. If that methodology were used on this patient, we would overestimate her medication
adherence. When a patient changes therapy during the course of treatment, any excess medication from the original
prescription is assumed to be discarded. While there may be overlap in medication coverage due to early refills, there
is no overlap because of switches to drug therapy.
The logic behind the new methodology in %PDC_Change accounts for early refills, similar to %PDC_Single, but it
also makes adjustments when new drugs from a different therapeutic class are introduced. %PDC_Change performs
a look-back (default is 180 days) from the start of the measurement period. The macro searches for existing
prescription fills to determine if a patient is new to therapy. If a patient has exiting medication fills that extend into the
measurement period, the applicable days of supply becomes part of the PDC calculation.
To calculate PDC for Patient B, the initial step is to align claims by fill date within the therapeutic class. Overlapping
days of supply are adjusted similar to the methodology described for %PDC_Single. Once individual medications are
aligned, the macro adjusts for overlaps between therapies. When the patient switches to a new therapy, any
remaining supply of the initial therapy is truncated and not counted towards the total days of therapy.

The flexibility built into this code allows the user to measure a fixed period of time before or after the index date or to
allow for various starting dates within a fixed period. To measure adherence for one year moving forward from a fixed
point, the variable IndexdateVar is set to the starting date (i.e.,01JAN2011), the status variable is set to POST, and
the ObserveWindow is set to 365. To measure adherence for the one-year period prior to the IndexdateVar, the
status variable is set to PRE. If the study has variable start dates for each patient, then the IndexdateVar can be
matched to an intervention date so that PDC is calculated based on each patient’s starting date.
**************************************************************************************
** Macro Parameters:
***
** input
- Input Dataset name
***
** output
- Output Dataset name
***
** IndexdateVar
- index date to calculate PDC
***
**
(can be a constant or at patient level)
***
** ObserveWindow
- days retrospect / followed to measure PDC ***
** WindowBeforeObserve - extra days used to evaluate leftover drugs ***
**
or to identify new to therapy status.
***
**
Default is 180 days.
***
** level
- Medispan GPI level. Default is at GPI6.
***
** pid
- Unique patient identifier
***
** filldt
- Prescription fill date
***
** dos
- Fill days of supply
***
** status
- Pre or Post period from IndexdateVar
***
**************************************************************************************
%macro PDC_Change(
input
=
, output
=
, indexdateVar
=
, ObserveWindow
=
, WindowBeforeObserve = 180
, level
= GPI6
, pid
=
, filldt
=
, dos
=
, status
=);
** Step 1: Extend study period to decide if a patient has
**;
** any leftover drugs or is new to therapy
**;
data _ClaimTmp;
set &input;
%if %UPCASE(&status) eq PRE %then
%do;
startdt
= &indexdateVar - &WindowBeforeObserve - &ObserveWindow;
truestart = &indexdateVar - &ObserveWindow;
enddt
= &indexdateVar - 1;
%end;
%else %do;
startdt
= &indexdateVar - &WindowBeforeObserve;

truestart = &indexdateVar;
enddt
= &indexdateVar + &ObserveWindow - 1;
%end;
if &filldt >= startdt and &filldt <= enddt;
format startdt enddt truestart mmddyy10.;
run;
proc sort data=_ClaimTmp;
by &pid &level &filldt;
run;
** Step 2: Align claims if overlap exists within therapy
** Rxs have been adjusted to adjstart/adjend/days
** on &level (generally on GPI6)
** Truncate Rxs fall outside the end of study period
data _ClaimTmp1;
set _ClaimTmp;
by &pid &level &filldt;
format adjStart adjEnd lastdate mmddyy10.;
retain lastdate;
adjStart = &filldt;
adjEnd
= &filldt + &dos - 1;
if first.&level then
lastdate = .;
else do;
** If overlap then adjust start/end dates else keep as is;
if adjStart <= lastdate then do;
adjStart = lastdate + 1;
adjEnd
= adjStart + &dos - 1;
end;
end;
** Overflow check;
if adjStart > enddt then adjStart = enddt;
if adjEnd
> enddt then adjEnd
= enddt;

**;
**;
**;
**;

if lastdate = enddt then days = 0;
else days = adjEnd - adjStart + 1;
** Save last ending date;
lastdate = adjEnd;
run;
** Step 3: Adjust claims when leftovers found in PRE observe window **;
data _ClaimTmp2;
set _ClaimTmp1;
by &pid &level &filldt;
** when 1: new to therapy / when 0: Existing Therapy;
retain newTherapy;
* To New: Denominator: start = first fill date, Numerator: days;
* To Old: Denominator: start = truestart
, Numerator: days readjusted;
if first.&level then do;
if adjStart >= truestart then newTherapy = 1;
else newTherapy = 0;
end;
if newTherapy = 0 then do;
if adjstart < truestart then do;
if adjend < truestart then delete;
else do;
adjstart = truestart;
days = adjEnd - truestart + 1;
end;
end;
end;
run;
proc sort data=_ClaimTmp2;
by &pid adjstart;
run;

** Step 4: Calculate PDC on patient disease level - covers therapy change **;
data _PTPDC;
set _ClaimTmp2;
by &pid adjstart;
retain patStart patEnd nSUMdrugdays nSUMEmptydays nTotalDays;
format patStart patEnd date9.;
if first.&pid then do;
nSUMdrugdays = 0;
nSUMEmptydays = 0;
** patStart and patEnd are used to store continuous drug covers;
patStart
= adjstart;
patEnd
= adjEnd;
** Can be used to verify calculation process;
if newTherapy = 0 then do;
nTotalDays
= enddt - min(adjstart,truestart) + 1;
nSUMdrugdays = patEnd - patStart + 1;
if patStart
> truestart then nSUMEmptydays = patStart - truestart;
end;
else do;
nTotalDays
= enddt - adjstart + 1;
nSUMdrugdays =patEnd - patStart + 1;
end;
end;
else do;
** Data Processing;
** only deal with new claims whose adjend is greater than patend;
if adjEnd > patEnd then do;
** When gap detected;
if adjStart > patEnd + 1 then do;
nSUMdrugdays
= nSUMdrugdays + (adjEnd
- adjStart + 1);
nSUMEmptydays = nSUMEmptydays + (adjstart - patEnd
- 1);
patStart
= adjStart;
patEnd
= adjEnd;
end;
else do;
** No gap detected then Accumulate nodrug days;
nSUMdrugdays
= nSUMdrugdays + adjEnd - patEnd;
patEnd
= adjEnd;
end;
end;
end;
** data output;
if last.&pid then do;
if patEnd < enddt then do;
** Accumulate nodrug days;
nSUMEmptydays =nSUMEmptydays +
enddt - patEnd;
end;
output;
end;
run;
** Step 5: Calculate PDC **;
Data &output(keep=&pid newTherapy nSUMdrugdays nSUMEmptydays nTotalDays PDC);
set _PTPDC;
PDC = nSUMdrugdays / nTotalDays;
run;
%mend PDC_Change;
%PDC_Change (
input=claims,
output=pdc,
indexdateVar= '01JAN2011'd,
ObserveWindow=365,
pid=patient_id,
filldt=fill_date,
dos=days_supply,
status=post);

Conclusion
The paper demonstrates methodologies for calculating PDC that can be used for patients on single or multiple drug
therapies. The included code covers adjustments for early refills and a method for avoiding double-counting
prescription coverage in the event a patient changes therapies during the measurement period. PDC can be
calculated at both the therapeutic class level and the patient disease level. Furthermore, these techniques are also
applicable to other medication utilization measures such as persistency.

